Pojačavač sa zajedničkim emitorom kao pojačavač snage

pretpostavljen napon između kolektora i emitora, "izlazni" napon

(%i1)
```
vce: (Vcc/2)*(1+sin(wt));
```

(%o1) $$\frac{Vcc\ (\sin(wt)+1)}{2}$$

(%i2)
```
wxplot2d(vce/Vcc,[wt,0,2*%pi]);
```

(%t2)


(%o2)

odgovarajuća struja kolektora

(%i3)
```
ic: (Vcc/(2*R))*(1-sin(wt));
```

(%o3) $$\frac{Vcc\ (1-\sin(wt))}{2\,R}$$

(%i4)
```
wxplot2d(ic/(Vcc/R),[wt,0,2*%pi]);
```

(%t4)


(%o4)

trenutna snaga disipacije na tranzistoru

(%i5)
```
pd: vce*ic;
```

(%o5) $$\frac{Vcc^2\ (1-\sin(wt))\ (\sin(wt)+1)}{4\,R}$$

(%i6)
```
pd: ratsimp(pd);
```

(%o6)
$$-\frac{Vcc^2 \sin(wt)^2 - Vcc^2}{4\,R}$$

(%i7)
```
wxplot2d(pd/(Vcc^2/R),[wt,0,2*%pi]);
```

(%t7)



(%o7)

srednja snaga disipacije na tranzistoru

(%i8)
```
Pd: 1/(2*%pi)*integrate(pd,wt,0,2*%pi);
```

(%o8)
$$\frac{Vcc^2}{8\,R}$$

napon na otporniku, KZN

(%i9)
```
vr: Vcc-vce;
```

(%o9)
$$Vcc - \frac{Vcc\,(\sin(wt)+1)}{2}$$

(%i10)
```
vr: ratsimp(vr);
```

(%o10)
$$-\frac{Vcc\sin(wt) - Vcc}{2}$$

(%i11)
```
wxplot2d(vr/Vcc,[wt,0,2*%pi]);
```

(%t11)



(%o11)

trenutna snaga na otporniku

```
(%i12) pr: vr*ic;
```

$$(\%o12) \quad -\frac{Vcc\ (1-\sin(wt))\ (Vcc\sin(wt)-Vcc)}{4\,R}$$

```
(%i13) pr: ratsimp(pr);
```

$$(\%o13) \quad \frac{Vcc^2\sin(wt)^2-2\,Vcc^2\sin(wt)+Vcc^2}{4\,R}$$

```
(%i14) wxplot2d(pr/(Vcc^2/R),[wt,0,2*%pi]);
```



(%t14)

(%o14)

srednja snaga na otporniku

```
(%i15) Pr: 1/(2*%pi)*integrate(pr,wt,0,2*%pi);
```

$$(\%o15) \quad \frac{3\,Vcc^2}{8\,R}$$

trenutna snaga koju izvor za napajanje ulaže u kolo

```
(%i16) pcc: Vcc*ic;
```

$$(\%o16) \quad \frac{Vcc^2\ (1-\sin(wt))}{2\,R}$$

```
(%i17) wxplot2d(pcc/(Vcc^2/R),[wt,0,2*%pi]);
```



(%t17)

(%o17)

srednja snaga koju izvor za napajanje ulaže u kolo

```
(%i18) Pcc: 1/(2*%pi)*integrate(pcc,wt,0,2*%pi);
```

(%o18) $$\frac{Vcc^2}{2\,R}$$

zakon o održanju energije na nivou trenutne snage

```
(%i19) pcc-pr-pd;
```

(%o19) $$-\frac{Vcc^2\sin\left(wt\right)^2-2\,Vcc^2\sin\left(wt\right)+Vcc^2}{4\,R}+\frac{Vcc^2\sin\left(wt\right)^2-Vcc^2}{4\,R}+\frac{Vcc^2\ \left(1-\sin\left(wt\right)\right)}{2\,R}$$

```
(%i20) ratsimp(%);
```

(%o20) 0

zakon o održanju energije na nivou srednje snage

```
(%i21) Pcc-Pd-Pr;
```

(%o21) 0